

LINGUAGGIO ASP (completo)

Creazione di una pagina ASP

Un file di Active Server Pages (ASP) è un file di testo con estensione asp contenente una qualsiasi combinazione dei seguenti elementi:

- Testo
- Tag HTML
- Script sul lato server

Per creare rapidamente un file asp è possibile rinominare i file HTML sostituendo l'estensione del nome di file htm o html esistente con l'estensione asp. Se il file in questione non contiene funzionalità ASP, il server tralascerà la fase di elaborazione degli script ASP e invierà il file al client rapidamente. Ciò consente una notevole flessibilità agli sviluppatori Web, in quanto essi possono assegnare un'estensione asp ai file anche se prevedono di aggiungere le funzionalità ASP solo in un secondo momento.

Per pubblicare un file asp sul Web, salvare il nuovo file in una directory virtuale sul sito Web, assicurandosi che per tale directory sia stata impostata l'autorizzazione di esecuzione o creazione di script. Aprire quindi il file dal browser digitando il relativo URL. Si ricordi che le pagine ASP devono essere elaborate dal server e non è quindi possibile richiedere un file asp digitando il relativo percorso fisico. Dopo che il file è stato caricato nel browser, si noterà che il server ha restituito una pagina HTML. A prima vista ciò potrebbe sembrare strano, ma si tenga presente che il server analizza ed esegue tutti gli script sul lato server di ASP prima dell'invio del file. L'utente riceverà sempre la pagina in formato HTML standard.

Per creare i file asp è possibile utilizzare qualsiasi editor di testo. A mano a mano che si acquisirà maggiore familiarità, potrebbe risultare più produttivo utilizzare un editor con un supporto avanzato per ASP, quale Microsoft® Visual InterDev™. Per ulteriori informazioni, visitare il [sito Web di Microsoft Visual InterDev](#) (informazioni in lingua inglese).

Aggiunta di comandi script sul lato server

Uno script sul lato server è costituito da una serie di istruzioni che consentono di inoltrare i comandi al server Web in ordine sequenziale. Chi ha già sviluppato siti Web avrà probabilmente familiarità con gli script sul lato client, eseguiti nel browser. Nei file asp gli script sono distinti dal testo e dai tag HTML tramite delimitatori. Un *delimitatore* è un carattere o una sequenza di caratteri che contrassegna l'inizio o la fine di un'unità. Nel caso del linguaggio HTML tali delimitatori sono i simboli di minore (<) e maggiore (>), che racchiudono i tag HTML.

ASP utilizza i delimitatori <% e %> per racchiudere i comandi script. È possibile inserire tra i delimitatori qualsiasi comando valido per il linguaggio di script in uso. Nell'esempio seguente viene illustrata una semplice pagina HTML contenente un comando script:

```
<HTML>
  <BODY>
    Ultimo aggiornamento della pagina: <%= Now() %>.
  </BODY>
</HTML>
```

La funzione **Now()** di VBScript restituisce la data e ora correnti. Quando il server Web elabora questa pagina, sostituisce <%= Now() %> con la data e l'ora correnti e restituisce la pagina al browser con questo risultato:

Ultimo aggiornamento della pagina: 29/01/99 14.20.00.

I comandi inseriti tra i delimitatori sono denominati *comandi script primari*. Tali comandi vengono elaborati utilizzando il linguaggio di script primario. Tutti i comandi utilizzati all'interno di delimitatori di script devono essere validi per il linguaggio di script primario. Per impostazione predefinita il linguaggio di script primario è VBScript, ma è possibile impostare un linguaggio predefinito diverso. Per informazioni, vedere [Utilizzo di linguaggi di script](#).

Se si ha già familiarità con gli script sul lato client, si conoscerà il tag HTML <SCRIPT> utilizzato per racchiudere i comandi script e le espressioni. Il tag <SCRIPT> consente inoltre di creare script sul lato server ogni volta che è necessario definire procedure in più linguaggi all'interno di un file asp. Per ulteriori informazioni, vedere [Utilizzo di linguaggi di script](#).

Combinazione di testo HTML e comandi script

Tra i delimitatori ASP è possibile racchiudere qualsiasi istruzione, espressione, procedura o operatore valido per il linguaggio di script primario in uso. Un'istruzione, in VBScript e altri linguaggi di script, è un'unità completa da un punto di vista sintattico che esprime un tipo di azione, dichiarazione o definizione. L'istruzione condizionale **If...Then...Else** riportata di seguito è un'istruzione VBScript comune.

```
<%
  Dim dtmHour
  dtmHour = Hour(Now())
  If dtmHour < 12 Then
    strGreeting = "Buongiorno!"
  Else
    strGreeting = "Salve!"
  End If
```

%>

```
<%= strGreeting %>
```

In base all'ora corrente, questo script assegna il valore "Buongiorno!" o "Salve!" alla variabile stringa `strGreeting`. L'istruzione `<%= strGreeting %>` invia al browser il valore corrente della variabile.

Un utente che visualizza questo script prima di mezzogiorno (in base al fuso orario del server Web) vedrà quindi la riga di testo seguente:

Buongiorno!

Un utente che visualizza questo script a mezzogiorno o dopo vedrà la riga di testo seguente:

Salve!

È possibile includere testo HTML tra le sezioni di un'istruzione. Lo script seguente, che utilizza un testo HTML all'interno di un'istruzione **If...Then...Else**, genera lo stesso risultato dello script dell'esempio precedente:

```
<%  
    Dim dtmHour  
    dtmHour = Hour(Now())  
    If dtmHour < 12 Then
```

```
%>
```

```
    Buongiorno!
```

```
<% Else %>
```

```
    Salve!
```

```
<% End If %>
```

Se la condizione è vera, ovvero se l'ora è precedente al mezzogiorno, il server Web invierà al browser il testo HTML che segue la condizione ("Buongiorno"). In caso contrario, invierà al browser il testo HTML che segue **Else** ("Salve!"). Questo tipo di combinazione di testo HTML e comandi script risulta utile per racchiudere in un'istruzione

If...Then...Else più righe di testo HTML. L'esempio precedente risulta invece maggiormente utile quando si desidera visualizzare un messaggio di benvenuto in più punti della pagina Web. È possibile impostare il valore della variabile una sola volta e visualizzarlo ripetutamente.

Aniché combinare il testo HTML con comandi script, è possibile restituire il testo HTML al browser dall'interno di un comando script. Per restituire il testo al browser, utilizzare l'oggetto ASP predefinito **Response**. L'esempio seguente genera lo stesso risultato degli script precedenti:

```
<%  
    Dim dtmHour  
  
    dtmHour = Hour(Now())  
  
    If dtmHour < 12 Then  
        Response.Write "Buongiorno!"  
    Else  
        Response.Write "Salve!"  
    End If
```

```
%>
```

Response.Write invia al browser il testo che segue. Utilizzare **Response.Write** all'interno di un'istruzione quando si desidera definire in modo dinamico il testo restituito al browser. Potrebbe ad esempio essere necessario creare una stringa contenente i valori di più variabili. Per ulteriori informazioni sull'oggetto **Response** e gli oggetti in generale, vedere [Utilizzo di componenti e oggetti](#) e [Invio di contenuto al browser](#). Per il momento si tenga semplicemente presente che esistono vari modi per inserire comandi script in una pagina HTML.

Tra i delimitatori ASP è possibile includere procedure scritte nel linguaggio di script primario. Per ulteriori informazioni, vedere [Utilizzo di linguaggi di script](#).

Oppure

Utilizzo di istruzioni ASP

ASP include istruzioni di output e di elaborazione che non fanno parte del linguaggio di script in uso.

L'istruzione di output `<%= espressione %>` visualizza il valore di un'espressione. Questa istruzione di output equivale all'utilizzo di **Response.Write** per la visualizzazione delle informazioni. L'espressione di output `<%= città %>` visualizza ad esempio Roma (il valore corrente della variabile) nel browser.

L'istruzione di elaborazione `<%@ parola chiave %>` fornisce ad ASP le informazioni necessarie per elaborare un file asp. L'istruzione seguente imposta ad esempio VBScript come linguaggio di script primario della pagina:

```
<%@ LANGUAGE=VBScript %>
```

L'istruzione di elaborazione deve essere inserita nella prima riga di un file asp. Se si desidera aggiungere più istruzioni a una pagina, le istruzioni dovranno essere racchiuse entro lo stesso delimitatore. Non inserire l'istruzione di elaborazione in un file incluso tramite l'istruzione **#include**. Per ulteriori informazioni, vedere [Inclusione di file](#).

Inserire uno spazio tra il simbolo (@) e la parola chiave. L'istruzione di elaborazione utilizza le parole chiavi seguenti:

- [LANGUAGE](#), imposta il linguaggio di script del file asp. Per informazioni, vedere [Utilizzo di linguaggi di script](#).

- [ENABLESESSIONSTATE](#), specifica se un file asp utilizza lo stato della sessione. Per informazioni, vedere [Gestione di sessioni](#).
- [CODEPAGE](#), imposta la tabella codici (la codifica dei caratteri) del file asp.
- [LCID](#), imposta l'identificatore delle impostazioni internazionali per il file.
- [TRANSACTION](#), specifica che il file asp verrà eseguito in un contesto transazionale. Per informazioni, vedere [Nozioni fondamentali sulle transazioni](#).

Importante In una singola istruzione possono essere inserite più parole chiave. Le coppie di parole chiave o di valori devono essere separate da uno spazio. Non inserire spazi accanto al segno di uguale (=).

Nell'esempio seguente vengono impostati sia il linguaggio di script sia la tabella codici:

```
<%@ LANGUAGE="JScript" CODEPAGE="932" %>
```

Spazi bianchi degli script

Se si utilizza VBScript o JScript come linguaggio di script primario, ASP rimuoverà gli spazi bianchi dai comandi. Se si utilizza un qualsiasi altro linguaggio di script, ASP conserverà gli spazi bianchi in modo che i linguaggi in cui la posizione e il rientro hanno un ruolo fondamentale vengano interpretati correttamente. Sono considerati spazi bianchi gli spazi, le tabulazioni, i ritorni a capo e gli avanzamenti di riga.

In VBScript e JScript è possibile utilizzare gli spazi bianchi dopo il delimitatore di apertura e prima del delimitatore di chiusura al fine di semplificare la lettura dei comandi. Le istruzioni seguenti sono tutte valide:

```
<% Color = "Verde" %>
```

```
<%Color="Verde"%>
```

```
<%
Color = "Verde"
%>
```

ASP rimuoverà gli spazi bianchi tra il delimitatore di chiusura di un'istruzione e il delimitatore di apertura dell'istruzione successiva. È consigliabile tuttavia utilizzare gli spazi per migliorare la leggibilità del codice. Se è necessario conservare gli spazi bianchi tra le istruzioni, ad esempio per visualizzare i valori delle variabili in una frase, utilizzare il carattere HTML che indica lo spazio unificatore (). Ad esempio:

```
<%
'Definisce due variabili con valori stringa.
strFirstName = "Luca"
strLastName = "Farace"
%>
```

```
<P>Questa pagina Web è personalizzata per "<%= strFirstName %>&nbsp;<%=
strLastName %>." </P>
```

Utilizzo di linguaggi di script

I linguaggi di programmazione quali Visual Basic, C++ e Java forniscono un accesso di basso livello alle risorse del computer e consentono di creare programmi complessi di grandi dimensioni. I linguaggi di script, invece, consentono di creare programmi con capacità limitate, denominati *script*, che eseguono funzioni del sito Web su un browser o un server Web. A differenza dei più complessi linguaggi di programmazione, i linguaggi di script vengono *interpretati*, il che significa che le istruzioni vengono eseguite in ordine sequenziale da un programma intermediario denominato interprete dei comandi. Nonostante l'interpretazione riduca l'efficienza di esecuzione, i linguaggi di script sono facili da apprendere e forniscono potenti funzionalità. Gli script possono essere incorporati nelle pagine HTML per formattarne il contenuto oppure possono essere utilizzati per implementare componenti COM che incapsulano la logica aziendale avanzata.

ASP consente agli sviluppatori Web di scrivere script eseguibili sul server in vari linguaggi di script. All'interno di un singolo file asp è infatti possibile utilizzare più linguaggi di script. Inoltre, poiché gli script vengono letti ed elaborati sul lato server, non è necessario che il browser che richiede il file asp supporti gli script.

È possibile utilizzare qualsiasi linguaggio di script per il quale è installato il modulo di script appropriato nel server Web. Un *modulo di script* è un programma che elabora i comandi scritti in un particolare linguaggio. Con ASP vengono forniti due moduli di script: Microsoft Visual Basic Scripting Edition (VBScript) e Microsoft JScript. È tuttavia possibile installare e utilizzare moduli di altri linguaggi di script, quali REXX, PERL e Python.

Gli sviluppatori Visual Basic potranno iniziare sin da subito a utilizzare VBScript, un sottoinsieme di Visual Basic. Gli sviluppatori Java, C o C++ noteranno similitudini con la sintassi JScript, sebbene JScript non sia direttamente correlato a Java o C.

Gli sviluppatori che utilizzano altri linguaggi di script, quali REXX, Perl o Python, potranno ottenere e installare il modulo di script appropriato in modo da poter applicare il linguaggio che già conoscono. Poiché ASP è un host di script COM, per utilizzare un linguaggio è necessario installare un modulo di script che sia conforme allo standard COM per la creazione di script e risieda come oggetto COM (Component Object Model) sul server Web.

Impostazione del linguaggio di script primario

Il *linguaggio di script primario* di ASP è il linguaggio utilizzato per elaborare i comandi tra i delimitatori <% e %>. Per impostazione predefinita il linguaggio di script primario è VBScript. È possibile utilizzare come linguaggio di script primario qualsiasi altro linguaggio di script per cui è disponibile un modulo di script. È inoltre possibile impostare il linguaggio di script primario per ogni singola pagina oppure per tutte le pagine di un'applicazione ASP.

Impostazione del linguaggio per un'applicazione

Per impostare il linguaggio di script primario per tutte le pagine di un'applicazione, impostare la proprietà **Linguaggio ASP predefinito** nella scheda **Opzioni applicazioni** dello snap-in Internet Information Services. Per ulteriori informazioni, vedere [Configurazione di applicazioni ASP](#).

Impostazione del linguaggio per una pagina

Per impostare il linguaggio di script primario per una singola pagina, aggiungere l'istruzione <%@ LANGUAGE %> all'inizio del file asp. La sintassi di tale istruzione è la seguente:

```
<%@ LANGUAGE=LinguaggioDiScript %>
```

dove *LinguaggioDiScript* è il linguaggio di script primario che si desidera impostare per quella particolare pagina. L'impostazione del linguaggio per una pagina ha priorità rispetto all'impostazione globale per tutte le pagine dell'applicazione.

Attenersi alle istruzioni per l'utilizzo di un'istruzione ASP. Per ulteriori informazioni, vedere [Creazione di una pagina ASP](#).

Nota Per utilizzare come linguaggio di script primario un linguaggio che non supporta la sintassi **Oggetto.Metodo**, è prima necessario creare la chiave del Registro di sistema **LanguageEngines**. Per ulteriori informazioni, vedere [Informazioni sul Registro di sistema](#).

Utilizzo di VBScript e JScript sul server

Quando si utilizza VBScript sul server con ASP, vengono disattivate due funzioni VBScript. Poiché gli script scritti con ASP vengono eseguiti sul server, le istruzioni VBScript che visualizzano gli elementi dell'interfaccia utente, **InputBox** e **MsgBox**, non sono supportate. Inoltre, negli script sul lato server non utilizzare le funzioni di VBScript **CreateObject** e **GetObject**, ma utilizzare invece **Server.CreateObject** in modo da consentire ad ASP di tenere traccia dell'istanza dell'oggetto. Gli oggetti creati tramite **CreateObject** o **GetObject** non possono accedere agli oggetti ASP predefiniti né prendere parte alle transazioni. Le uniche due eccezioni a questa regola riguardano l'utilizzo degli oggetti Admin di IIS e l'utilizzo dei moniker Java. Per ulteriori informazioni, vedere [Utilizzo degli oggetti Admin di IIS](#) e [Creazione di un oggetto da una classe Java](#).

Per un elenco e la descrizione di tutti gli operatori, le funzioni, le istruzioni, gli oggetti, le proprietà e i metodi di JScript, vedere la documentazione relativa a VBScript e JScript disponibile nel sito Web Microsoft [Windows Script Technologies](#).

Inserimento di commenti

Poiché in ASP l'elaborazione di tutti gli script viene eseguita sul lato server, non è necessario includere tag HTML di commento per nascondere gli script ai browser che non supportano gli script, come viene spesso fatto con gli script sul lato client. Tutti i comandi ASP vengono elaborati prima dell'invio del contenuto al browser. È possibile utilizzare i commenti HTML per aggiungere osservazioni a una pagina HTML. Tali commenti verranno restituiti al browser e potranno essere letti dall'utente visualizzando il codice HTML d'origine.

Commenti VBScript

In VBScript sono supportati i commenti introdotti dall'apostrofo. A differenza dei commenti HTML, questo tipo di commento viene rimosso durante l'elaborazione dello script e non è inviato al browser.

```
<%  
    'Questa riga e le due seguenti sono commenti.  
    'La funzione PrintTable stampa tutti  
    'gli elementi di una matrice.  
    PrintTable MyArray()  
%>
```

Non è possibile inserire un commento in un'espressione di output. La prima riga riportata di seguito verrà ad esempio eseguita, mentre la seconda non produrrà alcun risultato perché inizia con <%=.

```
<% i = i +1 'Questa istruzione incrementa i. (Script corretto.) %>
```

```
<%= name 'Questa istruzione stampa il nome della variabile. (Script errato.)  
%>
```

Commenti JScript

In JScript sono supportati i caratteri di commento //. Tali caratteri dovrebbero essere utilizzati in ciascuna riga del commento.

```
<%  
    var x  
    x = new Date()  
    // Questa riga invia la data corrente al browser,  
    // convertita in stringa.  
    Response.Write(x.toString())  
%>
```

Response.Write(x.toString())

Distinzione tra maiuscole e minuscole In VBScript non viene fatta distinzione tra maiuscole e minuscole. Per fare riferimento all'oggetto **Request** è ad esempio possibile specificare indifferentemente **Request** o **request**. Uno svantaggio della mancata distinzione tra maiuscole e minuscole è dato dal fatto che non si può utilizzare la formattazione maiuscola o minuscola per distinguere i nomi delle variabili. Non è ad esempio possibile creare due variabili distinte denominate Colore e colore.

In JScript viene invece fatta distinzione tra maiuscole e minuscole. Quando si utilizzano le parole chiave JScript negli script, è necessario digitare la parola chiave esattamente come indicato nella pagina di riferimento corrispondente. Se ad esempio si specifica **date** anziché **Date**, verrà generato un errore. La combinazione di maiuscole o minuscole indicata nella presente documentazione per gli oggetti ASP predefiniti è valida per i comandi JScript.

Utilizzo di variabili e costanti

Una *variabile* è un percorso di memorizzazione denominato presente nella memoria del computer e contenente dati, quali un numero o una stringa di testo. I dati contenuti in una variabile vengono denominati *valore* della variabile. Le variabili consentono di memorizzare, recuperare e modificare i valori tramite l'utilizzo di nomi che identificano la funzione svolta dallo script.

Dichiarazione e denominazione di variabili

Per la dichiarazione e la denominazione delle variabili, attenersi alle regole e alle linee guida del linguaggio di script in uso. Anche se non è necessario dichiarare una variabile prima di utilizzarla, è consigliabile eseguire abitualmente tale operazione per prevenire l'insorgenza di errori. *Dichiarare* una variabile significa indicare al modulo di script che una variabile con un nome particolare esiste, in modo che sia possibile inserire dei riferimenti alla variabile in qualsiasi punto di uno script.

VBScript

Sebbene VBScript non richieda la dichiarazione delle variabili, è consigliabile dichiarare tutte le variabili prima di utilizzarle. Per dichiarare una variabile in VBScript, utilizzare l'istruzione **Dim**, **Public** o **Private**. Ad esempio:

```
<% Dim UserName %>
```

Per rendere obbligatoria la dichiarazione esplicita delle variabili tramite le istruzioni **Dim**, **Private**, **Public** e **ReDim**, è possibile utilizzare l'istruzione **Option Explicit** di VBScript in un file asp. L'istruzione **Option Explicit** deve essere inserita dopo tutte le istruzioni ASP e prima di qualsiasi testo HTML o comando script. Questa istruzione ha effetto solo sui comandi ASP scritti in VBScript, non sui comandi JScript.

```
<% Option Explicit %>
```

```
<HTML>
```

```
<%
```

```
    Dim strUserName
```

```
    Public lngAccountNumber
```

```
%>
```

```
.  
. .  
.
```

Per ulteriori informazioni su questi comandi, vedere la documentazione relativa a VBScript, disponibile nel sito Web Microsoft [Windows Script Technologies](#).

JScript

Sebbene JScript in genere non richieda la dichiarazione delle variabili, è consigliabile dichiarare tutte le variabili prima di utilizzarle. Per dichiarare una variabile, utilizzare l'istruzione **var**. Ad esempio:

```
<% var UserName %>
```

In genere in JScript la dichiarazione di una variabile è richiesta solo quando è necessario distinguere una variabile interna a una funzione da una variabile *globale* utilizzata all'esterno della funzione. In tali casi, se non si distinguono le due variabili, JScript supporrà che si stia facendo riferimento solo alla variabile globale. Per ulteriori informazioni sull'elemento **var**, vedere la documentazione relativa a JScript, disponibile nel sito Web Microsoft [Windows Script Technologies](#).

Ambito delle variabili

L'*ambito*, o ciclo di vita, di una variabile determina quali comandi script hanno accesso alla variabile. Una variabile dichiarata all'interno di una procedura ha *ambito locale*, il che significa che la variabile viene creata e distrutta a ogni esecuzione della procedura. La variabile non sarà accessibile per tutti gli elementi esterni alla procedura. Una variabile dichiarata all'esterno di una procedura ha *ambito globale*, il che significa che il relativo valore è accessibile e modificabile da parte di qualsiasi comando script di una pagina ASP.

Nota La limitazione dell'ambito di una variabile a una procedura consente di ottenere un miglioramento delle prestazioni.

Se si ricorre alla dichiarazione delle variabili, è possibile assegnare uno stesso nome a una variabile locale e a una variabile globale. La modifica del valore di una delle due variabili non avrà effetto sul valore dell'altra. Se non si ricorre alla dichiarazione delle variabili, invece, si potrebbe modificare inavvertitamente il valore di una variabile globale. I comandi script seguenti restituiscono ad esempio il valore 1 nonostante esistano due variabili denominate Y:

```
<%
```

```

Option Explicit
Dim Y

Y = 1

SetLocalVariable

Response.Write Y

Sub SetLocalVariable
    Dim Y
    Y = 2
End Sub
%>

```

I comandi script seguenti, invece, restituiscono il valore 2 perché le variabili non sono dichiarate in modo esplicito. Quando la chiamata della procedura imposta Y su 2, il modulo di script suppone che lo scopo della procedura fosse quello di modificare la variabile globale:

```

<%
Option Explicit
Dim Y = 1

SetLocalVariable

Response.Write Y

Sub SetLocalVariable
    Y = 2
End Sub
%>

```

Al fine di evitare problemi, dichiarare sempre tutte le variabili in modo esplicito. La dichiarazione delle variabili assume particolare importanza quando si utilizza l'istruzione **#include** per includere uno o più file all'interno del file asp. Lo script incluso è contenuto in un file separato, ma viene utilizzato come se facesse parte del file di destinazione. Se non si ricorre alla dichiarazione delle variabili, è facile dimenticarsi che è necessario assegnare nomi diversi alle variabili dello script principale e dello script incluso.

Assegnazione dell'ambito a livello di sessione o applicazione

Le variabili globali sono accessibili in un singolo file asp. Se si desidera rendere accessibile una variabile anche all'esterno di una singola pagina, assegnare alla variabile l'ambito a livello di sessione o applicazione. Le variabili con ambito a livello di sessione sono disponibili per tutte le pagine di un'applicazione ASP richieste da un particolare utente. Le variabili con ambito a livello di applicazione sono disponibili per tutte le pagine di un'applicazione ASP richieste da qualsiasi utente. Le prime consentono di memorizzare informazioni relative a un singolo utente, quali le preferenze oppure il nome o l'identificazione dell'utente. Le seconde consentono di memorizzare informazioni relative a tutti gli utenti di una particolare applicazione, quali il messaggio visualizzato da un'applicazione all'accesso o i valori generali richiesti dall'applicazione.

ASP fornisce due oggetti predefiniti nei quali è possibile memorizzare le variabili, ovvero l'oggetto **Session** e l'oggetto **Application**.

È inoltre possibile creare istanze di oggetti con ambito a livello di sessione o applicazione. Per ulteriori informazioni, vedere [Impostazione dell'ambito degli oggetti](#).

Ambito a livello di sessione

Per assegnare a una variabile l'ambito a livello di sessione, memorizzarla nell'oggetto **Session** specificando un valore per la voce denominata dell'oggetto. I comandi seguenti memorizzano ad esempio due nuove variabili nell'oggetto **Session**:

```

<%
    Session("Nome") = "Giovanni"
    Session("Cognome") = "Giudici"
%>

```

Per recuperare informazioni dall'oggetto **Session**, accedere alla voce con nome utilizzando l'istruzione di output (<%=) o **Response.Write**. Nell'esempio seguente viene utilizzata l'istruzione di output per visualizzare il valore corrente di Session("Nome"):

```

Benvenuto <%= Session("Nome") %>

```

È possibile memorizzare le preferenze dell'utente nell'oggetto **Session**, quindi accedere a tali preferenze per determinare la pagina da restituire all'utente. È ad esempio possibile consentire a un utente di specificare una versione di solo testo del contenuto della prima pagina dell'applicazione e applicare tale scelta a tutte le pagine successive che l'utente visiterà in questa applicazione.

```

<%
    strScreenResolution = Session("ScreenResolution")
    If strScreenResolution = "Low" Then
%>
    Versione di solo testo della pagina.
<% Else %>
    Versione multimediale della pagina.
<% End If %>

```

Nota Se all'interno di uno script si fa più volte riferimento a una variabile con ambito a livello di sessione, è consigliabile ricorrere all'assegnazione a una variabile locale, come nell'esempio precedente, per ottenere un miglioramento delle prestazioni.

Ambito a livello di applicazione

Per assegnare a una variabile l'ambito a livello di applicazione, memorizzarla nell'oggetto **Application** specificando un valore per una voce denominata dell'oggetto. Il comando seguente memorizza ad esempio il messaggio visualizzato da un'applicazione nell'oggetto **Application**:

```
<% Application("Saluto") = "Benvenuti nel reparto vendite." %>
```

Per recuperare informazioni dall'oggetto **Application**, utilizzare l'istruzione di output (<%=) di ASP o

Response.Write per accedere alla voce denominata da tutte le pagine successive dell'applicazione. Nell'esempio seguente viene utilizzata l'istruzione di output per visualizzare il valore di Application("Saluto"):

```
<%= Application("Saluto") %>
```

Anche in questo caso, se lo script fa più volte riferimento a una variabile con ambito a livello di applicazione, è consigliabile ricorrere all'assegnazione a una variabile locale per ottenere un miglioramento delle prestazioni.

Utilizzo di costanti

Una *costante* è un nome utilizzato in sostituzione di un numero o una stringa. Alcuni componenti di base forniti con ASP, quali gli oggetti dati ActiveX (ADO, ActiveX Data Objects), definiscono costanti che è possibile utilizzare negli script. Un componente può dichiarare costanti in una *libreria di tipi del componente*, ossia in un file contenente informazioni sugli oggetti e i tipi supportati da un componente COM. Dopo aver dichiarato una libreria dei tipi nel file asp, è possibile utilizzare le costanti definite in tutti gli script dello stesso file asp. Analogamente, è possibile dichiarare una libreria dei tipi nel file Global.asa per utilizzare le costanti definite in tutti i file asp di un'applicazione. Per dichiarare una libreria dei tipi, utilizzare il tag <METADATA> nel file asp o Global.asa in uso. Ad esempio, per dichiarare la libreria dei tipi ADO, è possibile specificare le istruzioni seguenti:

```
<!--METADATA NAME="Microsoft ActiveX Data Objects 2.5 Library" TYPE="TypeLib"
UUID="{00000205-0000-0010-8000-00AA006D2EA4}"-->
```

In alternativa, per fare riferimento alla libreria dei tipi, anziché l'identificatore univoco universale (UUID, Universal Unique Identifier), è possibile specificare il percorso del file:

```
<!-- METADATA TYPE="typelib" FILE="c:\programmi\file
comuni\system\ado\msado15.dll"-->
```

È quindi possibile utilizzare le costanti ADO nel file asp in cui è stata dichiarata la libreria dei tipi o in un file asp residente in un'applicazione che contiene un file Global.asa con la dichiarazione della libreria dei tipi ADO.

Nell'esempio seguente adOpenKeyset e adLockOptimistic sono costanti ADO:

```

<%
    'Crea e apre l'oggetto Recordset.
    Set rstCustomerList = Server.CreateObject("ADODB.Recordset")

    rstCustomerList.ActiveConnection = cnnPubs
    rstCustomerList.CursorType = adOpenKeyset
    rstCustomerList.LockType = adLockOptimistic
%>

```

Nella tabella seguente sono elencate alcune librerie comunemente utilizzate con i relativi UUID:

Libreria dei tipi	UUID
Microsoft ActiveX Data Objects 2.5 Library	{00000205-0000-0010-8000-00AA006D2EA4}
Microsoft CDO 1.2 Library for Windows 2000 Server	{0E064ADD-9D99-11D0-ABE5-00AA0064D470}
MSWC Advertisement Rotator Object Library	{090ACFA1-1580-11D1-8AC0-00C0F00910F9}
MSWC IIS Log Object Library	{B758F2F9-A3D6-11D1-8B9C-080009DCC2FA}

Per informazioni sul tag <METADATA>, vedere [Dichiarazioni TypeLibrary](#).

Nelle versioni precedenti di ASP alcuni componenti fornivano le definizioni delle costanti all'interno di file che dovevano essere inclusi in tutti i file asp che utilizzavano le costanti. L'utilizzo dell'istruzione **#include** per l'inclusione delle definizioni delle costanti è supportata anche nell'attuale versione, ma le librerie dei tipi in genere risultano più

facili da utilizzare e semplificano l'aggiornamento degli script. Nelle versioni future di ASP i componenti potrebbero non fornire file di definizione delle costanti.

Nota L'utilizzo del tag <METADATA> anziché dell'istruzione **#include** può consentire un miglioramento delle prestazioni dell'applicazione Web.

È possibile definire costanti personalizzate. In VBScript utilizzare l'istruzione **Const**. In JScript è possibile utilizzare l'istruzione **var** per assegnare un valore costante alla variabile. Se si desidera utilizzare le costanti personalizzate in più file asp, inserire le definizioni in un file separato e includerle in tutti i file asp che utilizzano tali costanti.

Interazione con script sul lato client

Utilizzando ASP per la generazione e la modifica di script sul lato client è possibile estendere l'efficacia di questo ambiente. È ad esempio possibile scrivere script sul lato server che assemblino gli script sul lato client in base a variabili specifiche del server, il tipo di browser dell'utente o i parametri delle richieste HTTP.

Integrando istruzioni di script sul lato server all'interno di script sul lato client (racchiusi tra tag HTML <SCRIPT>), come indicato nel modello di esempio seguente, è possibile inizializzare e modificare gli script sul lato client in modo dinamico al momento della richiesta:

```
<SCRIPT LANGUAGE="VBScript">
<!--
```

```
variabile = <%=valore definito dal server %>
```

```
.
.
.
```

```
script sul lato client
```

```
<% script sul lato server utilizzato per generare un'istruzione sul lato
client %>
```

```
script sul lato client
```

```
.
.
.
-->
```

```
</SCRIPT>
```

L'incorporamento di tali funzionalità consente utili e interessanti applicazioni. Di seguito è ad esempio riportato un semplice script sul lato server (scritto in VBScript) che modifica uno script sul lato client (scritto in JScript):

```
<%
Dim dtmTime, strServerName, strServerSoftware, intGreeting

dtmTime = Time()
strServerName = Request.ServerVariables("SERVER_NAME")
strServerSoftware = Request.ServerVariables("SERVER_SOFTWARE")
```

```
'Genera un numero casuale.
Randomize
intGreeting = int(rnd * 3)
%>
```

```
<SCRIPT LANGUAGE="JScript">
<!--
```

```
//Richiama la funzione per visualizzare il saluto
showIntroMsg()
```

```
function showIntroMsg()
{
    switch(<%= intGreeting %>)
    {
        case 0:
```

```

    msg = "Il server Web <%= strServerName%> che esegue <%=
strServerSoftware %>."
    break
case 1:
    msg = "Server Web <%= strServerName%>. Ora locale <%= dtmTime %>."
    break
case 2:
    msg = "Il server esegue <%= strServerSoftware %>."
    break
}

document.write(msg)

}

-->
</SCRIPT>

```

Gli script di questo tipo possono essere estesi, ad esempio, per configurare un database sul lato client o uno script di personalizzazione DHTML. Questa tecnica innovativa consente inoltre di ridurre le connessioni ripetute client/server e l'elaborazione a carico del server.

Elaborazione dell'input dell'utente

L'oggetto [Request](#) di ASP consente di creare script semplici ma potenti per la raccolta e l'elaborazione dei dati raccolti tramite moduli HTML. In questo argomento, non solo si imparerà a creare script semplici per l'elaborazione dei moduli, ma si apprenderanno anche le tecniche necessarie per la convalida delle informazioni dei moduli, sia sul server Web che nel browser dell'utente.

Informazioni sui moduli HTML

I moduli HTML, il metodo più diffuso per la raccolta di informazioni basate sul Web, sono costituiti da tag HTML speciali opportunamente disposti che visualizzano elementi dell'interfaccia utente in una pagina Web. Le caselle di testo, i pulsanti e le caselle di controllo sono tutti esempi di elementi che permettono agli utenti di interagire con una pagina Web e inviare informazioni a un server Web.

I tag HTML seguenti generano ad esempio un modulo in cui un utente può immettere il nome, il cognome e l'età e contenente un pulsante per l'invio delle informazioni a un server Web. Questo modulo contiene anche un tag di input nascosto, non visualizzato dal browser, che consente di trasmettere ulteriori informazioni a un server Web.

```

<FORM METHOD="Post" ACTION="Profile.asp">
<INPUT TYPE="Text" NAME="FirstName">
<INPUT TYPE="Text" NAME="LastName">
<INPUT TYPE="Text" NAME="Age">
<INPUT TYPE="Hidden" NAME="UserStatus" VALUE="Nuovo">
<INPUT TYPE="Submit" VALUE="Invia">
</FORM>

```

Un'analisi completa dei tag dei moduli HTML esula dagli obiettivi di questo argomento, ma esistono numerose fonti di informazioni a cui è possibile attingere per imparare a creare moduli HTML utili ed efficaci. La funzione del browser per la visualizzazione del codice sorgente consente ad esempio di esaminare i moduli HTML creati in altri siti Web. Per informazioni sui metodi di utilizzo più innovativi dei moduli HTML con altre tecnologie per Internet, visitare il sito Web [MSDN Online](#) (informazioni in lingua inglese).

Elaborazione degli input dei moduli con ASP

Dopo aver creato un modulo HTML, è necessario elaborare l'input dell'utente, ossia inviare le informazioni a un file asp per l'analisi e la modifica. Esaminare ancora una volta il codice HTML dell'esempio precedente. Si noti che l'attributo ACTION del tag <FORM> fa riferimento a un file denominato Profile.asp. Quando l'utente invia le informazioni in formato HTML, il browser utilizza il metodo POST per inviare tali informazioni a un file asp sul server, in questo caso il file Profile.asp. Il file asp può contenere script che elaborano le informazioni e interagiscono con altri script, componenti COM o risorse, quali un database.

Esistono tre modi principali per raccogliere le informazioni dai moduli HTML tramite ASP:

- Un file htm statico può contenere un modulo che invia i relativi valori a un file asp.
- Un file asp può creare un modulo che invia le informazioni a un altro file asp.
- Un file asp può creare un modulo che invia le informazioni a se stesso, ossia al file asp contenente il modulo.

I primi due metodi indicati funzionano allo stesso modo dei moduli che interagiscono con programmi di altri server Web, fatta eccezione per il fatto che in ASP le operazioni di raccolta ed elaborazione delle informazioni dei moduli sono notevolmente semplificate. Il terzo metodo è particolarmente utile e verrà descritto nella sezione [Convalida dell'input dei moduli](#).

Raccolta dell'input dei moduli

L'oggetto **Request** di ASP fornisce due insiemi che semplificano la raccolta delle informazioni dei moduli inviate sotto forma di richiesta di URL.

L'insieme **QueryString**

L'insieme **QueryString** recupera i valori dei moduli passati al server Web sotto forma di testo posizionato dopo un punto di domanda nell'URL richiesto. I valori del modulo possono essere aggiunti all'URL richiesto utilizzando il metodo HTTP GET oppure manualmente.

Se ad esempio il modulo dell'esempio precedente utilizzasse il metodo GET (METHOD="GET") e l'utente digitasse *Giovanni, Giudici e 30*, al server verrebbe inviata la richiesta di URL seguente:

```
http://Reskit/Workshop1/Painting/Profile.asp?FirstName=Giovanni&LastName=Giudici&Age=30&UserStatus=New
```

Profile.asp potrebbe contenere il seguente script per l'elaborazione dei moduli:

```
Buongiorno <%= Request.QueryString("FirstName") %> <%=
```

```
Request.QueryString("LastName") %>.
```

```
Tu hai <%= Request.QueryString("Age") %> anni.
```

```
<%  
  If Request.QueryString("UserStatus") = "New" Then  
    Response.Write "Questa è la tua prima visita al sito."  
  End if
```

```
%>
```

In tal caso il server Web restituirebbe il testo seguente al browser dell'utente:

```
Buongiorno Giovanni Giudici. Tu hai 30 anni. Questa è la tua prima visita al sito.
```

L'insieme **QueryString** offre inoltre un parametro facoltativo che consente di accedere a uno dei vari valori visualizzati nella richiesta di URL (utilizzando il metodo GET). È inoltre possibile utilizzare la proprietà **Count** per contare il numero di volte per cui viene visualizzato un tipo specifico di valore.

Un modulo contenente una casella di riepilogo con vari elementi potrebbe ad esempio generare la richiesta seguente:

```
http://Reskit/CibiOrganici/elenco.asp?Food=Mele&Food=Olive&Food=Pane
```

Per contare più valori si potrebbe utilizzare il comando seguente:

```
Request.QueryString("Food").Count
```

Per visualizzare vari tipi di valori, Elenco.asp potrebbe contenere lo script seguente:

```
<%  
  lngTotal = Request.QueryString("Food").Count  
  For i = 1 To lngTotal  
    Response.Write Request.QueryString("Food")(i) & "<BR>"  
  Next
```

```
%>
```

Lo script precedente visualizzerebbe:

```
Mele
```

```
Olive
```

```
Pane
```

È inoltre possibile visualizzare l'intero elenco di valori sotto forma di stringa delimitata da virgole utilizzando:

```
<% Response.Write Request.QueryString("Item") %>
```

In tal caso verrebbe visualizzata la seguente stringa:

```
Mele, Olive, Pane
```

Raccolta di moduli

Quando si utilizza il metodo HTTP GET per passare a un server Web valori di moduli lunghi e complessi, si corre il rischio di perdere parte delle informazioni. Alcuni server Web tendono a limitare le dimensioni della stringa della query dell'URL, in modo che i valori di moduli lunghi passati con il metodo GET possano essere troncati. Se è necessario inviare una grande quantità di informazioni da un modulo a un server Web, utilizzare il metodo HTTP POST. Questo metodo, che invia i dati dei moduli nel corpo della richiesta HTTP, è in grado di trasmettere a un server un numero illimitato di caratteri. Per recuperare i valori inviati con il metodo POST, è possibile utilizzare l'insieme **Form** dell'oggetto **Request** di ASP.

L'insieme **Form** memorizza i valori in maniera analoga all'insieme **QueryString**. Se, ad esempio, un utente ha compilato un modulo specificando un lungo elenco di nomi, per recuperare i nomi è possibile utilizzare lo script seguente:

```
<%  
  lngTotal = Request.Form("Food").Count  
  For i = 1 To lngTotal  
    Response.Write Request.Form("Food")(i) & "<BR>"  
  Next
```

```
%>
```

Convalida dell'input dei moduli

Un modulo Web ben progettato spesso include uno script sul lato client che convalida l'input dell'utente prima dell'invio delle informazioni al server. Gli *script di convalida* sono ad esempio in grado di verificare se l'utente ha immesso un numero valido o se una casella di testo è stata lasciata vuota. Se, ad esempio, il sito Web contiene un modulo che consente agli utenti di calcolare il tasso di rendimento di un investimento, sarà necessario verificare che l'utente abbia inserito i valori numerici o testuali nei campi appropriati per evitare che vengano inviate al server informazioni non valide.

In genere è preferibile eseguire tutte le operazioni di convalida possibili sul lato client. Oltre a informare gli utenti più rapidamente degli errori di input, la convalida sul lato client favorisce tempi di risposta migliori, riduce il carico di lavoro dei server e rende disponibile larghezza di banda per altre applicazioni.

Lo script sul lato client seguente convalida l'input dell'utente, verificando che il numero di conto immesso dall'utente sia effettivamente un numero, prima di inviare le informazioni al server:

```
<SCRIPT LANGUAGE="JScript">
```

```
function CheckNumber()
{
  if (isNumeric(document.UserForm.AcctNo.value))
    return true
  else
  {
    alert("Specificare un numero di conto valido.")
    return false
  }
}
```

```
//Funzione per determinare se il valore del modulo è un numero.
```

```
//Nota: il metodo JScript isNaN rappresenta il metodo più elegante per determinare se un
```

```
//valore non è un numero. Tuttavia, il metodo non è supportato in alcuni browser meno recenti.
```

```
function isNumeric(str)
```

```
{
  for (var i=0; i < str.length; i++)
  {
    var ch = str.substring(i, i+1)
    if( ch < "0" || ch > "9" || str.length == null)
    {
      return false
    }
  }
  return true
}
</SCRIPT>
```

```
<FORM METHOD="Get" ACTION="balance.asp" NAME="UserForm" ONSUBMIT="return
CheckNumber()">
```

```
<INPUT TYPE="Text" NAME="AcctNo">
<INPUT TYPE="Submit" VALUE="Invia">
```

```
</FORM>
```

Se tuttavia la convalida dei moduli richiede l'accesso a database, può essere utile ricorrere alla convalida sul lato server. Un sistema particolarmente vantaggioso per l'esecuzione di operazioni di convalida sul lato server consiste nel creare un modulo che invii le informazioni a se stesso. In tal caso il file asp contiene il modulo HTML che recupera l'input dell'utente. Si ricorda che per interagire con script sul lato client e codice HTML è possibile utilizzare ASP. Per ulteriori informazioni, vedere [Interazione con script sul lato client](#). L'input viene restituito allo stesso file, che convalida quindi le informazioni e provvede a informare l'utente in caso di input non valido.

Questo metodo di elaborazione e convalida dell'input dell'utente consente un notevole miglioramento della semplicità di utilizzo e dei tempi di risposta dei moduli basati sul Web. Inserendo ad esempio le informazioni sugli errori accanto al campo del modulo in cui sono state immesse le informazioni non valide, l'utente potrà identificare l'origine

dell'errore con maggiore facilità. (In genere i moduli basati sul Web inoltrano le richieste a una pagina Web separata contenente informazioni sugli errori. Gli utenti che non sono in grado di interpretare immediatamente queste informazioni potrebbero scoraggiarsi.)

Lo script seguente determina ad esempio se un utente ha immesso un numero di conto valido inviando le informazioni a se stesso (il file Verify.asp) e richiamando una funzione definita dall'utente che esegue una query in un database:

```
<%
    strAcct = Request.Form("Account")
    If Not AccountValid(strAcct) Then
        ErrMsg = "<FONT COLOR=Red>Il numero di conto specificato non è
valido.</FONT>"
    Else
        Elabora l'input dell'utente
        .
        .
        .
        Server.Transfer("Complete.asp")
    End If

    Function AccountValid(strAcct)
        Posizione di una chiamata a uno script di connettività al database oppure
a un metodo del componente.
    End Function
%>
```

```
<FORM METHOD="Post" ACTION="Verify.asp">
Account Number: <INPUT TYPE="Text" NAME="Account"> <%= ErrMsg %> <BR>
<INPUT TYPE="Submit">
</FORM>
```

In questo esempio lo script è posizionato in un file denominato Verify.asp, lo stesso file contenente il modulo HTML. Lo script invia le informazioni a se stesso specificando Verify.asp nell'attributo ACTION.

Importante Se si utilizza JScript per la convalida sul lato server, assicurarsi di inserire una coppia di parentesi vuote dopo l'elemento dell'insieme **Request (QueryString o Form)** durante l'assegnazione dell'insieme a una variabile locale. Se non si specificano le parentesi, l'insieme restituirà un oggetto anziché una stringa. Lo script seguente indica come assegnare correttamente le variabili in JScript:

```
<%
    var Name = Request.Form("Name") ();
    var Password = Request.Form("Password") ();

    if(Name > "")
    {
        if(Name == Password)
            Response.Write("Il nome e la password sono identici.")
        else
            Response.Write("Il nome e la password sono diversi.");
    }
%>
```

VBScript presenta lo stesso comportamento quando l'insieme contiene più valori separati da virgole o indicizzabili. Questo significa che sia in VBScript che in JScript, oltre a inserire una coppia di parentesi vuote dopo l'elemento dell'insieme **Request**, è necessario specificare l'indice del valore desiderato. La riga di JScript seguente restituisce ad esempio solo il primo valore di un elemento di un modulo:

```
var Name = Request.Form("Name") (1);
```